# Maintainer: Alad Wenter <alad@mailbox.org>
# Contributor: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Matthias-Christian Ott <matthias.christian.at.tiscali.de>

pkgname=diffstat
pkgver=1.65
pkgrel=1
pkgdesc="Display a histogram of diff changes"
arch=('loong64' 'x86_64')
url="https://invisible-island.net/diffstat"
depends=('glibc')
license=('MIT')
source=("https://invisible-island.net/archives/$pkgname/$pkgname-$pkgver.tgz"
        "https://invisible-island.net/archives/$pkgname/$pkgname-$pkgver.tgz.asc")
sha256sums=('8cf27424926debcd85921755c3915559288244bd103f62d740dc6b83b557a28a'
            'SKIP')
validpgpkeys=('19882D92DDA4C400C22C0D56CC2AF4472167BE03') # Thomas Dickey <dickey@invisible-island.net>"

build () {
  cd "$pkgname-$pkgver"

  ./configure --prefix=/usr \
              --mandir=/usr/share/man
  make
}

package() {
  cd "$pkgname-$pkgver"

  make DESTDIR="$pkgdir/" install
  install -Dm644 package/debian/copyright "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
